

第3章 磁気流体力学方程式に対する近似リーマン解法

三好隆博 (広島大学)

この章では、磁気流体力学 (MHD) 方程式に対する代表的な衝撃波捕獲法である近似リーマン解法について解説する。

3.1 基礎方程式

MHD 方程式はプラズマの巨視的なダイナミクスを記述する基礎方程式である。特にここでは、電気抵抗率をゼロとした“理想”MHD 方程式を考える。

3.1.1 MHD 方程式

MHD 方程式は、外力としてローレンツ力を作用させた流体方程式と磁場の時間発展方程式の組み合わせによって与えられる。無次元化した抵抗性 MHD 方程式は、

$$\frac{\partial \rho}{\partial t} + \nabla \cdot (\rho \mathbf{v}) = 0 \quad : \quad \text{連続の式} \quad (3.1)$$

$$\rho \frac{d\mathbf{v}}{dt} = -\nabla p + \mathbf{j} \times \mathbf{B} \quad : \quad \text{運動方程式} \quad (3.2)$$

$$\frac{d}{dt} \left(\frac{p}{\rho^\gamma} \right) = \frac{(\gamma - 1) \eta j^2}{\rho^\gamma} \quad : \quad \text{断熱の式} \quad (3.3)$$

$$\frac{\partial \mathbf{B}}{\partial t} = -\nabla \times \mathbf{E} \quad : \quad \text{誘導方程式} \quad (3.4)$$

$$\mathbf{E} = -\mathbf{v} \times \mathbf{B} + \eta \mathbf{j} \quad : \quad \text{オームの法則} \quad (3.5)$$

$$\mathbf{j} = \nabla \times \mathbf{B} \quad : \quad \text{アンペールの法則} \quad (3.6)$$

$$\nabla \cdot \mathbf{B} = 0 \quad : \quad \text{磁場に関するガウスの法則} \quad (3.7)$$

となる。ここで、 ρ 、 $\mathbf{v}$ 、 p 、 $\mathbf{B}$ 、 $\mathbf{E}$ 、 $\mathbf{j}$ 、 γ 、 η は、密度、速度、圧力、磁場、電場、電流、比熱比、電気抵抗率をそれぞれ表す。ラグランジュ微分は、 $d/dt = \partial/\partial t + \mathbf{v} \cdot \nabla$ である。(3.7)は磁束の保存を表わしており、(3.4)の初期条件を与える。また、(3.5)、(3.6)は見かけ上消去できる。したがって、最終的に MHD 方程式は未知数 8 個の連立非線形偏微分方程式となる。特に電場が誘導電場のみで与えられるとき、すなわち (3.5) の η がゼロのとき、散逸のない理想 MHD 方程式となる。

理想 MHD 方程式 (3.1)-(3.7) は、以下のとおり保存形式で書き直すことができる。

$$\frac{\partial \mathbf{U}}{\partial t} + \nabla \cdot \mathbf{F} = 0, \quad (3.8)$$

$$\mathbf{U} = \begin{pmatrix} \rho \\ \rho \mathbf{v} \\ \mathbf{B} \\ e \end{pmatrix}, \quad \mathbf{F} = \begin{pmatrix} \rho \mathbf{v} \\ \rho \mathbf{v} \mathbf{v} + p_T \mathbf{I} - \mathbf{B} \mathbf{B} \\ \mathbf{v} \mathbf{B} - \mathbf{B} \mathbf{v} \\ (e + p_T) \mathbf{v} - \mathbf{B} (\mathbf{v} \cdot \mathbf{B}) \end{pmatrix}. \quad (3.9)$$

ここで、 $\mathbf{I}$ は単位行列を表す。全エネルギー密度 e と全圧力 p_T は、

$$e = \frac{p}{\gamma - 1} + \frac{\rho |\mathbf{v}|^2}{2} + \frac{|\mathbf{B}|^2}{2}, \quad (3.10)$$

$$p_T = p + \frac{|\mathbf{B}|^2}{2}, \quad (3.11)$$

により関係付けられる。理想 MHD 方程式のヤコビアン $\mathbf{A}$ 、

$$\mathbf{A} = \frac{\partial \mathbf{F}}{\partial \mathbf{U}} \quad (3.12)$$

は独立な実固有値 (特性速度) を持つ¹。したがって、理想 MHD 方程式は双曲型保存則系である。

3.1.2 1次元理想 MHD 方程式

この後の議論のため、改めて1次元の理想 MHD 方程式を書き下す。

$$\frac{\partial \mathbf{U}}{\partial t} + \frac{\partial \mathbf{F}}{\partial x} = 0, \quad (3.13)$$

$$\mathbf{U} = \begin{pmatrix} \rho \\ \rho u \\ \rho v \\ \rho w \\ B_y \\ B_z \\ e \end{pmatrix}, \quad \mathbf{F} = \begin{pmatrix} \rho u \\ \rho u^2 + p_T - B_x^2 \\ \rho v u - B_x B_y \\ \rho w u - B_x B_z \\ B_y u - B_x v \\ B_z u - B_x w \\ (e + p_T) u - B_x (u B_x + v B_y + w B_z) \end{pmatrix}. \quad (3.14)$$

ここで、 $\mathbf{v} = (u, v, w)$ 、 $\mathbf{B} = (B_x, B_y, B_z)$ とした。また、

$$p_T = p + \frac{1}{2} (B_x^2 + B_y^2 + B_z^2). \quad (3.15)$$

$$p = (\gamma - 1) \left[e - \frac{1}{2} \rho (u^2 + v^2 + w^2) - \frac{1}{2} (B_x^2 + B_y^2 + B_z^2) \right]. \quad (3.16)$$

¹ただし、固有値が縮退し得るため、厳密には固有値は完全独立ではない。

ただし、ガウスの法則 (3.7) から、1次元問題では、 B_x は時間・空間的に一定となる。

ヤコビアン $\mathbf{A}$ の固有値 λ_j ($j=1\dots 7$) は、(3.14) から、2つのアルヴェン波、2つの速進磁気音波、2つの遅進磁気音波、1つのエントロピー波として、

$$\lambda_1 = u - c_f, \lambda_2 = u - c_a, \lambda_3 = u - c_s, \lambda_4 = u, \lambda_5 = u + c_s, \lambda_6 = u + c_a, \lambda_7 = u + c_f \quad (3.17)$$

が得られる。ここで、

$$\begin{aligned} c_a &= |b_x|, \quad c_{f/s} = \sqrt{\frac{1}{2} \left[a^2 + b^2 \pm \sqrt{(a^2 + b^2)^2 - 4a^2 b_x^2} \right]}, \\ a &= \sqrt{\frac{\gamma p}{\rho}}, \quad b_x = \frac{B_x}{\sqrt{\rho}}, \quad b_y = \frac{B_y}{\sqrt{\rho}}, \quad b_z = \frac{B_z}{\sqrt{\rho}}, \quad b^2 = b_x^2 + b_y^2 + b_z^2. \end{aligned} \quad (3.18)$$

これから、

$$\lambda_1 \leq \lambda_2 \leq \lambda_3 \leq \lambda_4 \leq \lambda_5 \leq \lambda_6 \leq \lambda_7 \quad (3.19)$$

の不等式が成り立ち、磁場の向きや強さによって固有値が縮退し得ることがわかる。

一般に双曲型保存則系は、弱解 (積分形の保存則を満たす解) として衝撃波解や不連続解を持つことが知られている。特に理想 MHD 方程式では、磁場の強さや向きによって不連続解は全く異なる性質を持つ。衝撃波の後面で磁場が強められる速進衝撃波、磁場が弱められる遅進衝撃波、衝撃波面がアルヴェン速度で伝播するアルヴェン衝撃波が衝撃波解として存在する。ただし、アルヴェン衝撃波では、

$$[\rho] = [p] = [B_y^2 + B_z^2] = 0, \quad \pm\sqrt{\rho}[v] = [B_y], \quad \pm\sqrt{\rho}[w] = [B_z], \quad (3.20)$$

の条件が成り立ち、衝撃波の前後で熱力学的量に変化がないので、回転不連続 (rotational discontinuity) とも呼ばれる。ここで、 $[\cdot]$ は不連続前後の物理量の差を表す。また、エントロピー波と共に伝播する不連続解として、

$$[B_y] = [B_z] = [v] = [w] = [p] = 0, \quad B_x \neq 0 \quad (3.21)$$

を満足する接触不連続 (contact discontinuity)、

$$\left[p + \frac{B_y^2 + B_z^2}{2} \right] = 0, \quad B_x = 0 \quad (3.22)$$

を満足する接線不連続 (tangential discontinuity) が存在する。この他にも MHD 方程式では、複合波 (compound wave) や過圧縮衝撃波 (overcompressible shock) など普通の衝撃波でない衝撃波 (非古典的衝撃波) の存在が指摘されている。

3.2 近似リーマン解法

双曲型保存則系で生ずる不連続解を数値的に安定に解くため、特性の理論に基づく風上法が活発に開発研究され、大きな成功を収めてきた。このような衝撃波 (不連続解) を鋭く捕らえることができる解法を衝撃波捕獲法と呼ぶ。特にオイラー方程式に対しては、近似リーマン解法や流束ベクトル分離法、さらには AUSM (Advection Upstream Splitting Method) 系スキームといった様々なタイプの衝撃波捕獲法が開発された。流束ベクトル分離法は、オイラー方程式の流束関数 $\mathbf{F}$ が保存変数 $\mathbf{U}$ の一次同次関数




図 3.1: 近似リーマン解法のアルゴリズム。Step 1. 物理量を区分的定数関数によって近似する。Step 2. 各格子境界でリーマン問題を近似的に解く。Step 3. 各格子内でリーマン問題の解を積分する。

$\mathbf{F} = \mathbf{A}\mathbf{U}$ であることを利用して、 $\mathbf{F}$ を正負の固有値を持つ流束関数 $\mathbf{F}^\pm$ に分離する方法である。一方、MHD 方程式は $\mathbf{F} \neq \mathbf{A}\mathbf{U}$ であり、 $\mathbf{F}^\pm$ の構築は困難である。AUSM 系スキームでは、流束関数を移流項と音波による擾乱項に分離し、それぞれを風上化する²。MHD 方程式ではアルヴェン波が存在するため、既存の AUSM 系スキームを素直に適用することはできない³。したがって、MHD 方程式に対する実践的な衝撃波捕獲法としては、近似リーマン解法が用いられることがほとんどである。

流束ベクトル分離法が格子内の特性速度の符号により流束分離を行うのに対し、近似リーマン解法では格子境界の特性速度の情報を用いて流束の風上化を行う。近似リーマン解法の手続きを図 3.1 に示す。初期条件として、物理量を各格子内で一定となるよう区分的定数関数で与える。このとき各格子境界において物理量が不連続になるため、初期値問題は各格子境界におけるリーマン問題（衝撃波管問題）に帰着する。最終的に、各格子内でリーマン問題の解を積分することによって、ある区分的定数関数として次の時刻の解が近似的に求められる。この手続きを繰り返すことにより、双曲型方程式を数値的に解くことができる。特に、各格子境界でリーマン問題を厳密に解く方法を Godunov 法と呼ぶ。

ここで、 $[x_{i-1/2}, x_{i+1/2}] \times [t^n, t^{n+1}]$ における双曲型保存則系の時空間保存、

$$\int_{x_{i-1/2}}^{x_{i+1/2}} \mathbf{U}(x, t^{n+1}) dx - \int_{x_{i-1/2}}^{x_{i+1/2}} \mathbf{U}(x, t^n) dx + \int_{t^n}^{t^{n+1}} \mathbf{F}(\mathbf{U}(x_{i+1/2}, t)) dt - \int_{t^n}^{t^{n+1}} \mathbf{F}(\mathbf{U}(x_{i-1/2}, t)) dt = 0 \quad (3.23)$$

を考える。リーマン問題の解を $\mathbf{U}(x/t; \mathbf{U}_i, \mathbf{U}_{i+1})$ とすると、(3.23) は、

$$\mathbf{U}_i^{n+1} - \mathbf{U}_i^n + \frac{\Delta t}{\Delta x} [\mathbf{F}_{i+1/2}^* - \mathbf{F}_{i-1/2}^*] = 0, \quad (3.24)$$

ただし、

$$\mathbf{F}_{i+1/2}^* = \mathbf{F}(\mathbf{U}(0; \mathbf{U}_i, \mathbf{U}_{i+1})), \quad (3.25)$$

となる (図 3.2)。ここで、 $\Delta x = x_{i+1/2} - x_{i-1/2}$ 、 $\Delta t = t^{n+1} - t^n$ 。数値流束 $\mathbf{F}_{i+1/2}^*$ は、 $[x_{i+1/2} +$

²発見的方法によって定式化される。

³発見的方法によって定式化できる、かもしれない。パラメータ空間は無限である …。




図 3.2: $[x_{i-1/2}, x_{i+1/2}] \times [t^n, t^{n+1}]$ における双曲型保存則系のリーマン問題。

$\lambda_1 \Delta t, x_{i+1/2}] \times [t^n, t^{n+1}]$ の時空間保存から、

$$\mathbf{F}_{i+1/2}^* = \mathbf{F}_i^n - \frac{1}{\Delta t} \int_{\lambda_1 \Delta t}^0 U(x/t^{n+1}; \mathbf{U}_i^n, \mathbf{U}_{i+1}^n) dx - \lambda_1 \mathbf{U}_i^n \quad (3.26)$$

と与えられる。ただし、 λ_1 は最小固有値 (最小の特性速度) で負とする。 $\lambda_1 > 0$ の場合、超音速問題となるため、 $\mathbf{F}_{i+1/2}^* = \mathbf{F}_i^n$ と風上側の流束で置き換えられる。保存則の対称性から、

$$\mathbf{F}_{i+1/2}^* = \mathbf{F}_{i+1}^n + \frac{1}{\Delta t} \int_0^{\lambda_m \Delta t} U(x/t^{n+1}; \mathbf{U}_i^n, \mathbf{U}_{i+1}^n) dx - \lambda_m \mathbf{U}_{i+1}^n \quad (3.27)$$

も成立し、(3.26) と一致する。ただし、 λ_m は最大固有値 (最大の特性速度) で正とする。

オイラー方程式や MHD 方程式は極めて高度な非線形システム方程式であり、リーマン問題の厳密解を求めるためにはニュートン法などの反復計算が必要になる。一方、Godunov 法では最終的に各格子内でリーマン問題の解を平均化するため、解の詳細な情報は失われる。そこで、リーマン問題の厳密解の代替となる性質のよい近似解の探索が重要となる。一般に、リーマン問題の近似解を用いて数値流束を評価する解法を近似リーマン解法、または Godunov 型解法と呼ぶ。

これ以降、格子境界左右の状態を示す添え字をそれぞれ L 、 R とする。

3.2.1 線形近似リーマン解法

$[x_i, x_{i+1}] \times [t^n, t^{n+1}]$ においてヤコビアン $\mathbf{A}$ を一定とする近似リーマン解法、つまり局所的に基礎方程式を線形化する近似法を線形近似リーマン解法と呼ぶ。

ヤコビアン $\mathbf{A}_{i+1/2}$ を、正または負の固有値のみを持つヤコビアン $\mathbf{A}_{i+1/2}^\pm$ に分離する。ここで $\mathbf{A}_{i+1/2}^\pm$ は、 $\mathbf{A}_{i+1/2}$ に対する固有値行列 $\boldsymbol{\Lambda}_{i+1/2}$ と右固有行列 $\mathbf{R}_{i+1/2}$ を利用して、

$$\mathbf{A}_{i+1/2}^\pm = \frac{\mathbf{R}_{i+1/2} (\boldsymbol{\Lambda}_{i+1/2} \pm |\boldsymbol{\Lambda}_{i+1/2}|) \mathbf{R}_{i+1/2}^{-1}}{2} \quad (3.28)$$

と与えられる。したがって、

$$\mathbf{A}_{i+1/2} = \mathbf{A}_{i+1/2}^+ + \mathbf{A}_{i+1/2}^-, \quad |\mathbf{A}_{i+1/2}| = \mathbf{A}_{i+1/2}^+ - \mathbf{A}_{i+1/2}^-, \quad (3.29)$$

ただし、

$$|\mathbf{A}_{i+1/2}| = \mathbf{R}_{i+1/2} |\mathbf{A}| \mathbf{R}_{i+1/2}^{-1} \quad (3.30)$$

である。右に進む波をまたぐジャンプ条件から、

$$\mathbf{F}_R - \mathbf{F}^* = \mathbf{A}_{i+1/2}^+ (\mathbf{U}_R - \mathbf{U}_L), \quad (3.31)$$

同様に左に進む波をまたぐジャンプ条件から、

$$\mathbf{F}^* - \mathbf{F}_L = \mathbf{A}_{i+1/2}^- (\mathbf{U}_R - \mathbf{U}_L), \quad (3.32)$$

が得られる。最終的に数値流束を対称化して表すと、

$$\mathbf{F}^* = \frac{\mathbf{F}_R + \mathbf{F}_L}{2} - \frac{1}{2} |\mathbf{A}_{i+1/2}| (\mathbf{U}_R - \mathbf{U}_L) \quad (3.33)$$

となる。一般に、この解法は線形近似リーマン解法と呼ばれる。流束の差 $\Delta \mathbf{F}$ を $\Delta \mathbf{F}^+ + \Delta \mathbf{F}^-$ と分離することから流束差分分離法とも呼ばれる。特に、ジャンプ条件が正しく評価されるよう、

$$\mathbf{F}_R - \mathbf{F}_L = \mathbf{A}_{i+1/2} (\mathbf{U}_R - \mathbf{U}_L) \quad (3.34)$$

を満足する $\mathbf{A}_{i+1/2}$ を与える解法は Roe 法⁴と呼ばれ、最もスタンダードな近似リーマン解法としてよく知られている。

ただし、実際に MHD 方程式に線形近似リーマン解法を適用するには様々な工夫が必要となる。MHD 方程式では固有値 (3.17) が縮退し得るため、固有ベクトルに特異性が出現する。そのため固有ベクトルの適切な再規格化が必要となる⁵。また、 $\gamma = 2$ 以外の場合、Roe 法に対するヤコビアン $\mathbf{A}_{1/2}$ を求めることは容易ではない⁶。

線形近似リーマン解法では全ての波が考慮されており、各種不連続解を正しく解像できる。一方で、膨張波 (希薄波) が無視されているため、密度および圧力の正值性が破れることが証明されている。特に MHD 方程式では、磁気エネルギーの寄与分のため、オイラー方程式に比べてさらに正值性の条件は非常に厳しく、計算困難な初期条件が広く存在する。

3.2.2 HLL 近似リーマン解法

HLL 型の近似リーマン解法⁷では、膨張波も含めて最も速い特性速度 $S_R (> 0)$ と最も遅い速度 $S_L (< 0)$ に囲まれた領域、 $[S_L \Delta t, S_R \Delta t] \times [0, \Delta t]$ における時空間保存則、

$$\frac{1}{\Delta t} \int_{S_L \Delta t}^{S_R \Delta t} \mathbf{U}(x, t) dx - S_R \mathbf{U}_R + S_L \mathbf{U}_L + \mathbf{F}_R - \mathbf{F}_L = 0 \quad (3.35)$$

を満足するよう近似解が求められる。

⁴(3.34) に加えて、 $\mathbf{A}_{1/2}$ が独立な実固有値を持ち、ヤコビアン $\mathbf{A}_{1/2}(\mathbf{U}, \mathbf{U}) = \mathbf{A}(\mathbf{U})$ を満足する必要がある。これらの性質は合わせて Property U と呼ばれる。

⁵ここでは参考文献をあげるにとどめる。Brio and Wu, JCP, **75** (1988) 400。

⁶ここでは参考文献をあげるにとどめる。Cargo and Gallice, JCP, **136** (1997) 446、Balsara, ApJS, **116** (1998) 119。

⁷HLL 法の提案者である Harten(TVD/ENO の提案者)、Lax (双曲型保存則系業界の創始者)、van Leer(MUSCL の提案者) の頭文字。




図 3.3: HLL 近似リーマン解法における (a) 近似解の模式図と (b) 数値流束評価のための積分経路。

特に HLL 法では、 S_R と S_L に囲まれた領域 (リーマンファン) においてリーマン問題の解が単一状態 U^* であると仮定する。(3.35) から、

$$(S_R - S_L)U^* - S_R U_R + S_L U_L + F_R - F_L = 0, \quad (3.36)$$

したがって、

$$U^* = \frac{S_R U_R - S_L U_L - F_R + F_L}{S_R - S_L} \quad (3.37)$$

が得られる (図 3.3(a))。続いて、 $[0, S_R \Delta t] \times [0, \Delta t]$ における時空間保存 (図 3.3(b)) から、

$$S_R U^* - S_R U + F_R - F^* = 0 \quad (3.38)$$

となる。ただし、数値流束 F^* は、

$$F^* = \frac{1}{\Delta t} \int_0^{\Delta t} F(0, t) dt \quad (3.39)$$

と定義される。(3.37) を (3.38) に代入すると、 F^* は最終的に、

$$F^* = \frac{S_R F_L - S_L F_R + S_R S_L (U_R - U_L)}{S_R - S_L} \quad (3.40)$$

となる。もちろん、 $[S_L \Delta t, 0] \times [0, \Delta t]$ の保存則からも (3.40) は得られる。HLL 法は非線形の近似解法であり、 $F^* \neq F(U^*)$ であることに注意する必要がある。

HLL 法の数値流束 (3.40) を評価するためには、 S_R と S_L の情報が必要である。MHD 方程式の特性速度は形式的には (3.17) と与えられる。しかし、厳密な S_R 、 S_L を得るためには、リーマン問題を厳密に解く必要がある。そこで、 S_R 、 S_L を過小評価しないよう近似的に、

$$S_R = \max(u_R + c_{f_R}, u_L + c_{f_L}), \quad S_L = \min(u_R - c_{f_R}, u_L - c_{f_L}) \quad (3.41)$$

または

$$S_R = \max(u_R, u_L) + \max(c_{f_R}, c_{f_L}), \quad S_L = \min(u_R, u_L) - \max(c_{f_R}, c_{f_L}) \quad (3.42)$$

のように与えられる。また、特に Roe 平均値 (3.34) を用いて、

$$S_R = \max(u^{Roe} + c_f^{Roe}, u_R + c_{f_R}), \quad S_L = \max(u_L - c_{f_L}, u^{Roe} - c_f^{Roe}) \quad (3.43)$$

とも評価される。 $S_R < 0$ 、 $S_L > 0$ の場合についても、(3.40) が片側の流束 $\mathbf{F}_R$ 、 $\mathbf{F}_L$ に帰着できるよう、 S_R 、 S_L は一般に、

$$S_R = \max(S_R, 0), \quad S_L = \min(S_L, 0) \quad (3.44)$$

と拡張される。

HLL 法は $\mathbf{U}$ や $\mathbf{F}$ の詳細によらず非常にシンプルであり、固有ベクトルの計算も不要で計算効率が極めて高い。また、オイラー方程式および MHD 方程式に対して正値性が理論的に証明されており、極めてロバストである。しかし、リーマンファン内を単一状態と仮定しているため、接触不連続や回転不連続などを捕らえることができず、低解像度で数値的散逸が強い。

3.2.3 HLLD 近似リーマン解法

続いて、接触不連続や回転不連続を解像できる高解像度 HLL 型近似リーマン解法、HLLD 法⁸を検討する。HLLD 法では、リーマンファンにおいて速度 u が一定と仮定する。この仮定から、リーマンファン内で加速や減速が生じないよう、全圧力 p_T が一定であることも要求される (図 3.4(b))。したがって、リーマンファン内は非圧縮状態となり、リーマン問題の解から遅進衝撃波は除外される。このことから、リーマン問題の近似解は、2つの速進衝撃波 S_R 、 S_L 、2つのアルヴェン波 S_R^* 、 S_L^* 、1つのエントロピー波 S_M で分割された4つの状態 (リーマンファンの左から $\mathbf{U}_L^*$ 、 $\mathbf{U}_L^{**}$ 、 $\mathbf{U}_R^{**}$ 、 $\mathbf{U}_R^*$ と表記) となる (図 3.4(a))。

リーマンファンにおける速度 u はエントロピー波の速度 S_M と一致する。ここで S_M は、時空間保存則 (3.37) を用いて、

$$\begin{aligned} S_M &= \frac{(\rho u)^*}{\rho^*} \\ &= \frac{(S_R - u_R)\rho_R u_R - (S_L - u_L)\rho_L u_L - p_{TR} + p_{TL}}{(S_R - u_R)\rho_R - (S_L - u_L)\rho_L} \end{aligned} \quad (3.45)$$

と与える。速度 u はリーマンファン内で一定なので、

$$u_L^* = u_L^{**} = u_R^{**} = u_R^* = S_M \quad (3.46)$$

となる。また全圧力 p_T も、

$$p_{TL}^* = p_{TL}^{**} = p_{TR}^{**} = p_{TR}^* = p_T^* \quad (3.47)$$

と一定である。

まず、速進衝撃波 S_α (α は R または L) に対するジャンプ条件、

$$S_\alpha \begin{pmatrix} \rho_\alpha^* \\ \rho_\alpha^* S_M \\ \rho_\alpha^* v_\alpha^* \\ \rho_\alpha^* w_\alpha^* \\ B_{y\alpha}^* \\ B_{z\alpha}^* \\ e_\alpha^* \end{pmatrix} - \begin{pmatrix} \rho_\alpha^* S_M \\ \rho_\alpha^* S_M^2 + p_T^* - B_x^2 \\ \rho_\alpha^* v_\alpha^* S_M - B_x B_{y\alpha}^* \\ \rho_\alpha^* w_\alpha^* S_M - B_x B_{z\alpha}^* \\ B_{y\alpha}^* S_M - B_x v_\alpha^* \\ B_{z\alpha}^* S_M - B_x w_\alpha^* \\ (e_\alpha^* + p_T^*) S_M - B_x (\mathbf{v}_\alpha^* \cdot \mathbf{B}_\alpha^*) \end{pmatrix} = S_\alpha \begin{pmatrix} \rho_\alpha \\ \rho_\alpha u_\alpha \\ \rho_\alpha v_\alpha \\ \rho_\alpha w_\alpha \\ B_{y\alpha} \\ B_{z\alpha} \\ e_\alpha \end{pmatrix} - \begin{pmatrix} \rho_\alpha u_\alpha \\ \rho_\alpha u_\alpha^2 + p_{T\alpha} - B_x^2 \\ \rho_\alpha v_\alpha u_\alpha - B_x B_{y\alpha} \\ \rho_\alpha w_\alpha u_\alpha - B_x B_{z\alpha} \\ B_{y\alpha} u_\alpha - B_x v_\alpha \\ B_{z\alpha} u_\alpha - B_x w_\alpha \\ (e_\alpha + p_{T\alpha}) u_\alpha - B_x (\mathbf{v}_\alpha \cdot \mathbf{B}_\alpha) \end{pmatrix}, \quad (3.48)$$

⁸HLLD 法の “D” は不連続 (discontinuities) の頭文字を表す。HLLC 法 (接触不連続 (contact discontinuity) を解像する HLL 型の近似リーマン解法) の上位の解法である。




図 3.4: HLLD 近似リーマン解法における近似解の模式図。

を考えよう。第 2 式から、

$$\begin{aligned}
 p_T^* &= p_{TL} + \rho_L(S_L - u_L)(S_M - u_L) \\
 &= p_{TR} + \rho_R(S_R - u_R)(S_M - u_R) \\
 &= \frac{(S_R - u_R)\rho_R p_{TL} - (S_L - u_L)\rho_L p_{TR} + \rho_L \rho_R (S_R - u_R)(S_L - u_L)(u_R - u_L)}{(S_R - u_R)\rho_R - (S_L - u_L)\rho_L}
 \end{aligned} \quad (3.49)$$

が得られる。続いて第 1 式から、

$$\rho_\alpha^* = \rho_\alpha \frac{S_\alpha - u_\alpha}{S_\alpha - S_M} \quad (3.50)$$

となる。第 3 式と第 5 式を連立させて、

$$v_\alpha^* = v_\alpha - B_x B_{y\alpha} \frac{S_M - u_\alpha}{\rho_\alpha (S_\alpha - u_\alpha)(S_\alpha - S_M) - B_x^2} \quad (3.51)$$

$$B_{y\alpha}^* = B_{y\alpha} \frac{\rho_\alpha (S_\alpha - u_\alpha)^2 - B_x^2}{\rho_\alpha (S_\alpha - u_\alpha)(S_\alpha - S_M) - B_x^2} \quad (3.52)$$

が得られる。同様に、第 4 式と第 6 式から、

$$w_\alpha^* = w_\alpha - B_x B_{z\alpha} \frac{S_M - u_\alpha}{\rho_\alpha (S_\alpha - u_\alpha)(S_\alpha - S_M) - B_x^2} \quad (3.53)$$

$$B_{z\alpha}^* = B_{z\alpha} \frac{\rho_\alpha (S_\alpha - u_\alpha)^2 - B_x^2}{\rho_\alpha (S_\alpha - u_\alpha)(S_\alpha - S_M) - B_x^2} \quad (3.54)$$

が得られる。最後に第7式から、

$$e_\alpha^* = \frac{(S_\alpha - u_\alpha)e_\alpha - p_{T\alpha}u_\alpha + p_T^*S_M + B_x(\mathbf{v}_\alpha \cdot \mathbf{B}_\alpha - \mathbf{v}_\alpha^* \cdot \mathbf{B}_\alpha^*)}{S_\alpha - S_M} \quad (3.55)$$

となる。

続いて、アルヴェン波 S_α^* に対するジャンプ条件を考える。 ρ のジャンプ条件、すなわち連続の式から明らかに、任意の S_α^* に対し、

$$\rho_\alpha^{**} = \rho_\alpha^* \quad (3.56)$$

が成り立つ (図 3.4(c))。したがって、アルヴェン波の速度は、

$$S_R^* = S_M + \frac{|B_x|}{\sqrt{\rho_R^*}}, \quad S_L^* = S_M - \frac{|B_x|}{\sqrt{\rho_L^*}} \quad (3.57)$$

と与えられる。ところが、(3.57) を用いると、 $\mathbf{v}_t = (0, v, w)$ 、 $\mathbf{B}_t = (0, B_y, B_z)$ のジャンプ条件、

$$S_\alpha^* \begin{pmatrix} \rho_\alpha^* v_\alpha^{**} \\ \rho_\alpha^* w_\alpha^{**} \\ B_{y\alpha}^{**} \\ B_{z\alpha}^{**} \end{pmatrix} - \begin{pmatrix} \rho_\alpha^* v_\alpha^{**} S_M - B_x B_{y\alpha}^{**} \\ \rho_\alpha^* w_\alpha^{**} S_M - B_x B_{z\alpha}^{**} \\ B_{y\alpha}^{**} S_M - B_x v_\alpha^{**} \\ B_{z\alpha}^{**} S_M - B_x w_\alpha^{**} \end{pmatrix} = S_\alpha^* \begin{pmatrix} \rho_\alpha^* v_\alpha^* \\ \rho_\alpha^* w_\alpha^* \\ B_{y\alpha}^* \\ B_{z\alpha}^* \end{pmatrix} - \begin{pmatrix} \rho_\alpha^* v_\alpha^* S_M - B_x B_{y\alpha}^* \\ \rho_\alpha^* w_\alpha^* S_M - B_x B_{z\alpha}^* \\ B_{y\alpha}^* S_M - B_x v_\alpha^* \\ B_{z\alpha}^* S_M - B_x w_\alpha^* \end{pmatrix}, \quad (3.58)$$

から、 $\mathbf{v}_{t\alpha}^{**}$ 、 $\mathbf{B}_{t\alpha}^{**}$ を解くことができない。一方、 S_M に対する $\mathbf{v}_t$ 、 $\mathbf{B}_t$ のジャンプ条件、

$$S_M \begin{pmatrix} \rho_L^* v_L^{**} \\ \rho_L^* w_L^{**} \\ B_{yL}^{**} \\ B_{zL}^{**} \end{pmatrix} - \begin{pmatrix} \rho_L^* v_L^{**} S_M - B_x B_{yL}^{**} \\ \rho_L^* w_L^{**} S_M - B_x B_{zL}^{**} \\ B_{yL}^{**} S_M - B_x v_L^{**} \\ B_{zL}^{**} S_M - B_x w_L^{**} \end{pmatrix} = S_M \begin{pmatrix} \rho_R^* v_R^{**} \\ \rho_R^* w_R^{**} \\ B_{yR}^{**} \\ B_{zR}^{**} \end{pmatrix} - \begin{pmatrix} \rho_R^* v_R^{**} S_M - B_x B_{yR}^{**} \\ \rho_R^* w_R^{**} S_M - B_x B_{zR}^{**} \\ B_{yR}^{**} S_M - B_x v_R^{**} \\ B_{zR}^{**} S_M - B_x w_R^{**} \end{pmatrix}, \quad (3.59)$$

から、

$$v_L^{**} = v_R^{**} \equiv v^{**} \quad (3.60)$$

$$w_L^{**} = w_R^{**} \equiv w^{**} \quad (3.61)$$

$$B_{yL}^{**} = B_{yR}^{**} \equiv B_y^{**} \quad (3.62)$$

$$B_{zL}^{**} = B_{zR}^{**} \equiv B_z^{**} \quad (3.63)$$

となる (図 3.4(d))。そこで、リーマンファンの時空間保存則 (3.35)、

$$(S_R - S_R^*)\mathbf{U}_R^* + (S_R^* - S_M)\mathbf{U}_R^{**} + (S_M - S_R^*)\mathbf{U}_L^{**} + (S_L^* - S_L)\mathbf{U}_L^* - S_R\mathbf{U}_R + S_L\mathbf{U}_L + \mathbf{F}_R - \mathbf{F}_L = 0 \quad (3.64)$$

を利用すると、

$$v^{**} = \frac{\sqrt{\rho_L^*}v_L^* + \sqrt{\rho_R^*}v_R^* + (B_{yR}^* - B_{yL}^*)\text{sgn}(B_x)}{\sqrt{\rho_L^*} + \sqrt{\rho_R^*}} \quad (3.65)$$

$$w^{**} = \frac{\sqrt{\rho_L^*}w_L^* + \sqrt{\rho_R^*}w_R^* + (B_{zR}^* - B_{zL}^*)\text{sgn}(B_x)}{\sqrt{\rho_L^*} + \sqrt{\rho_R^*}} \quad (3.66)$$

$$B_y^{**} = \frac{\sqrt{\rho_L^*} B_{yR}^* + \sqrt{\rho_R^*} B_{yL}^* + \sqrt{\rho_L^* \rho_R^*} (v_R^* - v_L^*) \text{sgn}(B_x)}{\sqrt{\rho_L^*} + \sqrt{\rho_R^*}} \quad (3.67)$$

$$B_z^{**} = \frac{\sqrt{\rho_L^*} B_{zR}^* + \sqrt{\rho_R^*} B_{zL}^* + \sqrt{\rho_L^* \rho_R^*} (w_R^* - w_L^*) \text{sgn}(B_x)}{\sqrt{\rho_L^*} + \sqrt{\rho_R^*}} \quad (3.68)$$

が得られる。ここで、 sgn は符号関数である。最後に S_α^* に対する e のジャンプ条件から、

$$e_R^{**} = e_R^* + \sqrt{\rho_R^*} (\mathbf{v}_R^* \cdot \mathbf{B}_R^* - \mathbf{v}^{**} \cdot \mathbf{B}^{**}) \text{sgn}(B_x) \quad (3.69)$$

$$e_L^{**} = e_L^* - \sqrt{\rho_L^*} (\mathbf{v}_L^* \cdot \mathbf{B}_L^* - \mathbf{v}^{**} \cdot \mathbf{B}^{**}) \text{sgn}(B_x) \quad (3.70)$$

が得られる。ただし、 $\mathbf{v}^{**} = (S_M, v^{**}, w^{**})$ 、 $\mathbf{B}^{**} = (B_x, B_y^{**}, B_z^{**})$ 。

以上、リーマンファンにおける 4 つの状態、 $\mathbf{U}_L^*$ 、 $\mathbf{U}_L^{**}$ 、 $\mathbf{U}_R^{**}$ 、 $\mathbf{U}_R^*$ が完全に代数的に得られた。

数値流束 $\mathbf{F}^*$ は、HLL 法と同様に、 $[0, S_R \Delta t] \times [0, \Delta t]$ または $[S_L \Delta t, 0] \times [0, \Delta t]$ における時空間保存則から求められる。例えば、 $S_L \leq 0 \leq S_L^*$ の場合、

$$(0 - S_L) \mathbf{U}_L^* - (0 - S_L) \mathbf{U}_L + \mathbf{F}^* - \mathbf{F}_L = 0 \quad (3.71)$$

から、 $S_L^* \leq 0 \leq S_M$ の場合、

$$(0 - S_L^*) \mathbf{U}_L^{**} + (S_L^* - S_L) \mathbf{U}_L^* - (0 - S_L) \mathbf{U}_L + \mathbf{F}^* - \mathbf{F}_L = 0 \quad (3.72)$$

から $\mathbf{F}^*$ が得られる。最終的に、

$$\mathbf{F}^* = \begin{cases} \mathbf{F}_L & \text{for } 0 \leq S_L \\ \mathbf{F}_L^* & \text{for } S_L \leq 0 \leq S_L^* \\ \mathbf{F}_L^{**} & \text{for } S_L^* \leq 0 \leq S_M \\ \mathbf{F}_R^{**} & \text{for } S_M \leq 0 \leq S_R^* \\ \mathbf{F}_R^* & \text{for } S_R^* \leq 0 \leq S_R \\ \mathbf{F}_R & \text{for } S_R \leq 0 \end{cases} \quad (3.73)$$

となる⁹。ただし、

$$\mathbf{F}_{R/L}^{*/**} = \mathbf{F} \left(\rho_{R/L}^{*/**}, S_M, v_{R/L}^{*/**}, w_{R/L}^{*/**}, B_x, B_{yR/L}^{*/**}, B_{zR/L}^{*/**}, e_{R/L}^{*/**}, p_T^* \right) \quad (3.74)$$

である。 S_R と S_L は、HLL 法と同様に、例えば、(3.41) から (3.44) のように評価する。特に磁場がゼロのとき、この解法はオイラー方程式に対する HLLC 法に帰着する。

HLL 法と異なり、HLLD 法では MHD 方程式の接触不連続や回転不連続、接線不連続を正確に解像できる。また、密度および圧力の正値性の保存性も理論的に保証される (付録 3.3.1)¹⁰。様々な数値実験の結果から、線形近似リーマン解法と同等の解像度を持ちつつ、線形近似リーマン解法に比べ計算効率が低いことも示されている。したがって、HLLD 法は、解像度、ロバストさ、計算効率の全てに優れた MHD 解法といえる。

⁹ジャンプ条件 $S[\mathbf{U}] - [\mathbf{F}] = 0$ を満足するように近似解を求めたので当然である。

¹⁰MHD 方程式に対する HLL 法の正値性保存性も合わせて証明できる。

3.3 付録

3.3.1 HLL 型近似リーマン解法の正值性の証明

MHD 方程式の物理的な解の集合 G 、

$$G = \left\{ \mathbf{U} \left| \rho > 0, p = (\gamma - 1) \left[e - \frac{\rho(u^2 + v^2 + w^2)}{2} - \frac{B_x^2 + B_y^2 + B_z^2}{2} \right] > 0 \right. \right\} \quad (3.75)$$

を考える。 G に含まれる $\mathbf{U}_1$ と $\mathbf{U}_2$ の重み付き平均値 $\mathbf{U}$ 、

$$\mathbf{U} = (1 - \theta)\mathbf{U}_1 + \theta\mathbf{U}_2 \quad (\text{ただし、} 0 \leq \theta \leq 1) \quad (3.76)$$

に対して、密度 ρ と圧力 p は、

$$\rho = (1 - \theta)\rho_1 + \theta\rho_2 \quad (3.77)$$

$$p = (1 - \theta)p_1 + \theta p_2 + \theta(1 - \theta)(\gamma - 1) \left(\frac{\rho_1\rho_2|\mathbf{v}_1 - \mathbf{v}_2|^2}{2\rho} + \frac{|\mathbf{B}_1 - \mathbf{B}_2|^2}{2} \right) \quad (3.78)$$

と表される。したがって、 $\rho > 0$ 、 $p > 0$ であり、 $\mathbf{U}$ も G に含まれることがわかる。

近似リーマン解法による数値解は、各格子境界におけるリーマン問題の近似解を各格子内で積分したものに等しい。一方、上で示されたとおり、 G に含まれる $\mathbf{U}$ の重み付き平均値は G に含まれる。したがって、近似リーマン解法の正值性の保存性を証明するためには、リーマン問題の近似解の正值性を調べればよいことになる。

HLLD 近似リーマン解法における $\mathbf{U}_R^*$ 、 $\mathbf{U}_R^{**}$ の正值性の条件、

$$\rho_R^* > 0, \quad (3.79)$$

$$\rho_R^{**} > 0, \quad (3.80)$$

$$e_R^* - \frac{1}{2}\rho_R^*|\mathbf{v}_R^*|^2 - \frac{1}{2}|\mathbf{B}_R^*|^2 > 0, \quad (3.81)$$

$$e_R^{**} - \frac{1}{2}\rho_R^{**}|\mathbf{v}_R^{**}|^2 - \frac{1}{2}|\mathbf{B}_R^{**}|^2 > 0, \quad (3.82)$$

について議論しよう。

ここで、次のような変数、

$$\xi \equiv S_R - u_R, \quad \eta \equiv S_R - S_M, \quad \zeta \equiv S_M - u_R \quad (3.83)$$

を導入する。 S_R が、例えば (3.41)–(3.43) のように、膨張波も含めた系最大の特性速度よりも小さくならないように与えられるので、 ξ と η は常に正である。一方、 ζ は正と負のどちらも取り得る。また、変数の定義から $\xi - \eta = \zeta$ である。

密度の正值性の条件 (3.79)、(3.80) は、(3.50)、(3.56) から、

$$\rho_R^* = \rho_R^{**} = \frac{\xi}{\eta}\rho_R > 0 \quad (3.84)$$

と常に満たされることがわかる。

$\mathbf{U}_R^*$ における圧力の正值性の条件 (3.81) を示すため、

$$\varphi \equiv \eta \left(e_R^* - \frac{1}{2} \rho_R^* |\mathbf{v}_R^*|^2 - \frac{1}{2} |\mathbf{B}_R^*|^2 \right) \quad (3.85)$$

の正值性を証明しよう。 φ を整理すると、

$$\begin{aligned} \varphi &= \xi e_R - p_{TR} u_R + p_T^* S_M + B_x (\mathbf{v}_R \cdot \mathbf{B}_R - \mathbf{v}_R^* \cdot \mathbf{B}_R^*) - \eta \frac{\rho_R^* |\mathbf{v}_R^*|^2}{2} - \eta \frac{|\mathbf{B}_R^*|^2}{2} \\ &= \xi \left(\frac{p_R}{\gamma - 1} + \frac{\rho_R |\mathbf{v}_R|^2}{2} + \frac{|\mathbf{B}_R|^2}{2} \right) - \left(p_R + \frac{|\mathbf{B}_R|^2}{2} \right) u_R + \left(p_R + \frac{|\mathbf{B}_R|^2}{2} + \rho_R \xi \zeta \right) S_M \\ &\quad + B_x \left(-B_x \zeta - \frac{\rho_R \xi \zeta}{\rho_R \xi \eta - B_x^2} (\mathbf{v}_{tR} \cdot \mathbf{B}_{tR}) + \frac{B_x \zeta}{\rho_R \xi \eta - B_x^2} |\mathbf{B}_{tR}|^2 + \frac{\rho_R B_x \xi \zeta^2}{(\rho_R \xi \eta - B_x^2)^2} |\mathbf{B}_{tR}|^2 \right) \\ &\quad - \xi \frac{\rho_R |\mathbf{v}_R^*|^2}{2} - (\xi - \zeta) \frac{|\mathbf{B}_R^*|^2}{2} \\ &= \frac{p_R \xi}{\gamma - 1} + p_R \zeta + \frac{|\mathbf{B}_R|^2}{2} \zeta + \rho_R S_M \xi \zeta \\ &\quad + \frac{\rho_R (|\mathbf{v}_R|^2 - |\mathbf{v}_R^*|^2)}{2} \xi + \frac{|\mathbf{B}_R|^2 - |\mathbf{B}_R^*|^2}{2} \xi + \frac{|\mathbf{B}_R^*|^2}{2} \zeta \\ &\quad - B_x^2 \zeta - \frac{\rho_R B_x (\mathbf{v}_{tR} \cdot \mathbf{B}_{tR})}{\rho_R \xi \eta - B_x^2} \xi \zeta + \frac{B_x^2 |\mathbf{B}_{tR}|^2}{\rho_R \xi \eta - B_x^2} \zeta + \frac{\rho_R B_x^2 |\mathbf{B}_{tR}|^2}{(\rho_R \xi \eta - B_x^2)^2} \xi \zeta^2 \\ &= \frac{p_R \xi}{\gamma - 1} + p_R \zeta + \rho_R S_M \xi \zeta + \frac{\rho_R (u_R^2 - S_M^2)}{2} \xi \\ &\quad + \frac{\rho_R (|\mathbf{v}_{tR}|^2 - |\mathbf{v}_{tR}^*|^2)}{2} \xi + \frac{|\mathbf{B}_{tR}|^2 - |\mathbf{B}_{tR}^*|^2}{2} \xi + \frac{|\mathbf{B}_{tR}|^2 + |\mathbf{B}_{tR}^*|^2}{2} \zeta \\ &\quad - \frac{\rho_R B_x (\mathbf{v}_{tR} \cdot \mathbf{B}_{tR})}{\rho_R \xi \eta - B_x^2} \xi \zeta + \frac{B_x^2 |\mathbf{B}_{tR}|^2}{\rho_R \xi \eta - B_x^2} \zeta + \frac{\rho_R B_x^2 |\mathbf{B}_{tR}|^2}{(\rho_R \xi \eta - B_x^2)^2} \xi \zeta^2 \\ &= \frac{p_R \xi}{\gamma - 1} + p_R \zeta + \frac{\rho_R \xi}{2} \zeta^2 \\ &\quad + \frac{\rho_R \xi}{2} \left(\frac{2 B_x (\mathbf{v}_{tR} \cdot \mathbf{B}_{tR})}{\rho_R \xi \eta - B_x^2} \zeta - \frac{B_x^2 |\mathbf{B}_{tR}|^2}{(\rho_R \xi \eta - B_x^2)^2} \zeta^2 \right) \\ &\quad + \frac{\xi}{2} \left(-\frac{2 \rho_R |\mathbf{B}_{tR}|^2}{\rho_R \xi \eta - B_x^2} \xi \zeta - \frac{\rho_R^2 |\mathbf{B}_{tR}|^2}{(\rho_R \xi \eta - B_x^2)^2} \xi^2 \zeta^2 \right) \\ &\quad + \frac{1}{2} \left(2 |\mathbf{B}_{tR}|^2 + \frac{2 \rho_R |\mathbf{B}_{tR}|^2}{\rho_R \xi \eta - B_x^2} \xi \zeta + \frac{\rho_R^2 |\mathbf{B}_{tR}|^2}{(\rho_R \xi \eta - B_x^2)^2} \xi^2 \zeta^2 \right) \zeta \\ &\quad - \frac{\rho_R B_x (\mathbf{v}_{tR} \cdot \mathbf{B}_{tR})}{\rho_R \xi \eta - B_x^2} \xi \zeta + \frac{B_x^2 |\mathbf{B}_{tR}|^2}{\rho_R \xi \eta - B_x^2} \zeta + \frac{\rho_R B_x^2 |\mathbf{B}_{tR}|^2}{(\rho_R \xi \eta - B_x^2)^2} \xi \zeta^2 \\ &= \frac{p_R \xi}{\gamma - 1} + p_R \zeta + \frac{\rho_R \xi}{2} \left(1 - \frac{|\mathbf{B}_{tR}|^2}{\rho_R \xi \eta - B_x^2} \right) \zeta^2 \end{aligned} \quad (3.86)$$

となる。ただし、 $\mathbf{v}_{tR} = (0, v_R, w_R)$ 、 $\mathbf{B}_{tR} = (0, B_{yR}, B_{zR})$ 。途中、(3.49)–(3.55)、

$$p_{TR}^* = p_{TR} + \rho_R \xi \zeta \quad (3.87)$$

$$\rho_R^* = \frac{\xi}{\eta} \rho_R \quad (3.88)$$

$$\mathbf{v}_{tR}^* = \mathbf{v}_{tR} - \frac{B_x \zeta}{\rho_R \xi \eta - B_x^2} \mathbf{B}_{tR} \quad (3.89)$$

$$\mathbf{B}_{tR}^* = \frac{\rho_R \xi^2 - B_x^2}{\rho_R \xi \eta - B_x^2} \mathbf{B}_{tR} = \left(1 + \frac{\rho_R \xi \zeta}{\rho_R \xi \eta - B_x^2}\right) \mathbf{B}_{tR} \quad (3.90)$$

$$e_R^* = \frac{\xi e_R - p_{TR} u_R + p_T^* S_M + B_x (\mathbf{v}_R \cdot \mathbf{B}_R - \mathbf{v}_R^* \cdot \mathbf{B}_R^*)}{\eta} \quad (3.91)$$

を利用した。ここで、 S_R が系最大の特性速度であることから、 $\xi \geq c_{fR}$ 、 $\eta \geq c_{fR}$ となるので、

$$\varphi' = \frac{p_R \xi}{\gamma - 1} + p_R \zeta + \frac{\rho_R \xi}{2} \left(1 - \frac{|\mathbf{B}_{tR}|^2}{\rho_R c_{fR}^2 - B_x^2}\right) \zeta^2 \leq \varphi \quad (3.92)$$

である。つまり、 φ' が正であれば φ は必ず正である。 φ' は ζ の二次方程式であり、2次の項の係数は、

$$\rho c_f^2 - |\mathbf{B}|^2 = \rho c_f^2 - B_x^2 - |\mathbf{B}_t|^2 > 0 \quad (3.93)$$

から正である¹¹。したがって、 φ' の判別式 D が負のとき、任意の ζ に対して φ' は正となる。つまり、

$$D(\varphi') = p_R^2 - \frac{2\rho_R p_R}{\gamma - 1} \left(1 - \frac{|\mathbf{B}_{tR}|^2}{\rho_R c_{fR}^2 - B_x^2}\right) \xi^2 < 0 \quad (3.94)$$

から、

$$\xi^2 > \frac{(\gamma - 1)p_R}{2\rho_R} \left(1 - \frac{|\mathbf{B}_{tR}|^2}{\rho_R c_{fR}^2 - B_x^2}\right)^{-1} = \frac{\gamma - 1}{2\gamma} c_{fR}^2 \quad (3.95)$$

のとき、正值性の条件 (3.81) は満たされる。ただし、導出過程で、

$$\rho c_s^2 - B_x^2 = -\frac{B_x^2 |\mathbf{B}_t|^2}{\rho c_f^2 - B_x^2}, \quad c_f^2 c_s^2 = \frac{\gamma p B_x^2}{\rho^2} \quad (3.96)$$

を利用した。最終的に正值性の条件は、

$$S_R > u_R + \sqrt{\frac{\gamma - 1}{2\gamma}} c_{fR} \quad (3.97)$$

となり、 S_R は明らかにこの条件を満たす。

続いて、 $\mathbf{U}_R^{**}$ における圧力の正值性の条件 (3.82) は、

$$\begin{aligned} p_R^{**} &= (\gamma - 1) \left(e_R^{**} - \frac{\rho_R^* |\mathbf{v}^{**}|^2}{2} - \frac{|\mathbf{B}^{**}|^2}{2} \right) \\ &= p_R^* + \frac{\gamma - 1}{2} \left(\left| \sqrt{\rho_R^*} \mathbf{v}_R^* + \mathbf{B}_R^* \operatorname{sgn}(B_x) \right|^2 - \left| \sqrt{\rho_R^*} \mathbf{v}^{**} + \mathbf{B}^{**} \operatorname{sgn}(B_x) \right|^2 \right) \\ &= p_R^* > 0 \end{aligned} \quad (3.98)$$

と示される。

¹¹ $\mathbf{B}_t = 0$ 、 $B_x^2 \geq \gamma p$ のとき、 $|\mathbf{B}_t|^2 / (\rho c_f^2 - B_x^2)$ は 0/0 となる。ただし、 $\mathbf{B}_t = 0$ では、(3.86) はオイラー方程式に対する HLLC 近似リーマン解法の正值性条件に一致する。

U_L^* , U_L^{**} の正值性の条件は、対称性から、

$$S_L < u_L - \sqrt{\frac{\gamma-1}{2\gamma}} c_{fL} \quad (3.99)$$

となり、 S_L は容易にこの条件を満たす。

以上、リーマン問題の近似解、 U_L^* , U_L^{**} , U_R^* , U_R^{**} の正值性が示された。したがって、HLLD 近似リーマン解法は正值性を保存する解法である。

また、HLL 近似リーマン解法における近似解は、HLLD 近似リーマン解法における近似解の重み付き平均値として与えられる。したがって、HLL 近似リーマン解法も正值性を保存する解法である。

3.3.2 HLLD 近似リーマン解法のサンプルプログラム

HLLD 近似リーマン解法のサンプルプログラム (Fortran90) を示す。

```
!-----
subroutine calc_flux_hlld(rol,vnl,vtl,vul,btl,bul,prl, &
                        ror,vnr,vtr,vur,btr,bur,prl, &
                        bnc, &
                        fro,fmn,fmt,fmu,fbt,fbu,fen)
!-----
real(DP), intent(IN)  :: rol,vnl,vtl,vul,btl,bul,prl
real(DP), intent(IN)  :: ror,vnr,vtr,vur,btr,bur,prl
real(DP), intent(IN)  :: bnc
real(DP), intent(OUT) :: fro,fmn,fmt,fmu,fbt,fbu,fen
real(DP)               :: bnc2,sgn
real(DP)               :: roli,pml,ptl,enl,vbl
real(DP)               :: rori,pmr,ptr,enr,vbr
real(DP)               :: cl2,cal2,cbl2,cfl2,cfl
real(DP)               :: cr2,car2,cb2,cfr2,cfr
real(DP)               :: sr,sl
real(DP)               :: slvl,srvr,rslvl,rsrvr,drsvi
real(DP)               :: vnc,ptc
real(DP)               :: slvc,rhdl,rhdli,rhnvl,rhnbl
real(DP)               :: srvc,rhdr,rhdri,rhnvr,rhnbr
real(DP)               :: ro2l,vt2l,vu2l,bt2l,bu2l,vb2l,en2l
real(DP)               :: ro2r,vt2r,vu2r,bt2r,bu2r,vb2r,en2r
real(DP)               :: rro2l,rro2r,rro2i
real(DP)               :: vt3m,vu3m,bt3m,bu3m,vb3m,en3l,en3r
real(DP)               :: rou,vtu,vuu,btu,uuu,enu
!.....
```

```

! Bn at the interface
!.....
  bnc2 = bnc**2
  sgn  = sign(1.0_DP,bnc)
!.....
! variables at the left-face
!.....
  roli = 1.0_DP/rol
  pml  = 0.5_DP*(btl**2+bul**2)
  ptl  = prl+pml
  enl  = gammam1i*prl+pml+0.5_DP*rol*(vnl**2+vtl**2+vul**2)
  vbl  = vtl*btl+vul*bul
!.....
! variables at the right-face
!.....
  rori = 1.0_DP/ror
  pmr  = 0.5_DP*(btr**2+bur**2)
  ptr  = prr+pmr
  enr  = gammam1i*prr+pmr+0.5_DP*ror*(vnr**2+vtr**2+vur**2)
  vbr  = vtr*btr+vur*bur
!.....
! maximum / minimum wave speeds
!.....
  cl2  = gamma*prl*roli
  cr2  = gamma*prr*rori
  cal2 = bnc2*roli
  car2 = bnc2*rori
  cbl2 = cl2+cal2+2.0_DP*pml*roli
  cbr2 = cr2+car2+2.0_DP*pmr*rori
  cfl2 = 0.5_DP*(cbl2+sqrt(abs(cbl2**2-4.0_DP*cl2*cal2)))
  cfr2 = 0.5_DP*(cbr2+sqrt(abs(cbr2**2-4.0_DP*cr2*car2)))
  cfl  = sqrt(cfl2)
  cfr  = sqrt(cfr2)
  sl   = min(0.0_DP,min(vnl,vnr)-max(cfl,cfr))
  sr   = max(0.0_DP,max(vnl,vnr)+max(cfl,cfr))
!.....
! HLL average of the normal velocity and the total pressure
!.....
  slvl = sl-vnl
  srvr = sr-vnr

```

```

rslvl = rol*slvl
rsrvr = ror*srvr
drsvi = 1.0_DP/(rsrvr-rslvl)
vnc   = (rsrvr*vnr-rslvl*vnl-ptr+ptl)*drsvi
ptc   = (rsrvr*ptl-rslvl*ptr+rsrvr*rslvl*(vnr-vnl))*drsvi
!.....
! variables of the outer sides in the Riemann fan
!.....
slvc  = sl-vnc
srvc  = sr-vnc
ro2l  = rslvl/slvc
ro2r  = rsrvr/srvc
rhd1  = rslvl*slvc-bnc2
rhdr  = rsrvr*srvc-bnc2
if(abs(rhd1) > eps) then
  rhdli = 1.0_DP/rhd1
  rhnvl = (vnl-vnc)*bnc
  rhnbl = rslvl*slvl-bnc2
  vt2l = vtl+rhnvl*rhdli*bt1
  vu2l = vul+rhnvl*rhdli*bul
  bt2l =      rhnbl*rhdli*bt1
  bu2l =      rhnbl*rhdli*bul
else
  vt2l = vtl
  vu2l = vul
  bt2l = bt1
  bu2l = bul
end if
if(abs(rhdr) > eps) then
  rhdri = 1.0_DP/rhdr
  rhnvr = (vnr-vnc)*bnc
  rhnbr = rsrvr*srvr-bnc2
  vt2r = vtr+rhnvr*rhdri*btr
  vu2r = vur+rhnvr*rhdri*bur
  bt2r =      rhnbr*rhdri*btr
  bu2r =      rhnbr*rhdri*bur
else
  vt2r = vtr
  vu2r = vur
  bt2r = btr

```

```

    bu2r = bur
end if
vb2l = vt2l*bt2l+vu2l*bu2l
vb2r = vt2r*bt2r+vu2r*bu2r
en2l = (slvl*enl-ptl*vn1+ptc*vnc+bnc*(vbl-vb2l))/slvc
en2r = (srvr*enr-ptr*vn1+ptc*vnc+bnc*(vbr-vb2r))/srvc
!.....
! variables of the inner sides in the Riemann fan
!.....
    rro2l = sqrt(ro2l)
    rro2r = sqrt(ro2r)
    rro2i = 1.0_DP/(rro2r+rro2l)
    vt3m = (rro2r*vt2r+rro2l*vt2l+                (bt2r-bt2l)*sgn)*rro2i
    vu3m = (rro2r*vu2r+rro2l*vu2l+                (bu2r-bu2l)*sgn)*rro2i
    bt3m = (rro2l*bt2r+rro2r*bt2l+rro2r*rro2l*(vt2r-vt2l)*sgn)*rro2i
    bu3m = (rro2l*bu2r+rro2r*bu2l+rro2r*rro2l*(vu2r-vu2l)*sgn)*rro2i
    vb3m = vt3m*bt3m+vu3m*bu3m
    en3l = en2l-rro2l*(vb2l-vb3m)*sgn
    en3r = en2r+rro2r*(vb2r-vb3m)*sgn
!.....
! variables at the interface
!.....
    if(vnc-abs(bnc)/rro2l > 0.0_DP) then
        rou = ro2l
        vtu = vt2l
        vu1 = vu2l
        btu = bt2l
        buu = bu2l
        enu = en2l
    else if((vnc-abs(bnc)/rro2l <= 0.0_DP).and.(vnc >= 0.0_DP))then
        rou = ro2l
        vtu = vt3m
        vu1 = vu3m
        btu = bt3m
        buu = bu3m
        enu = en3l
    else if((vnc < 0.0_DP).and.(vnc+abs(bnc)/rro2r >= 0.0_DP))then
        rou = ro2r
        vtu = vt3m
        vu1 = vu3m

```

```

    btu = bt3m
    buu = bu3m
    enu = en3r
else
    rou = ro2r
    vtu = vt2r
    vuu = vu2r
    btu = bt2r
    buu = bu2r
    enu = en2r
end if
!.....
! HLLD fluxes
!.....
    fro = rou*vnc
    fmn = rou*vnc*vnc+ptc-bnc2*0.5_DP
    fmt = rou*vtu*vnc-bnc*btu
    fmu = rou*vuu*vnc-bnc*buu
    fbt = btu*vnc-bnc*vtu
    fbu = buu*vnc-bnc*vuu
    fen = (enu+ptc)*vnc-bnc*(vtu*btu+vuu*buu)
!-----
end subroutine calc_flux_hlld
!-----

```

◇ ◇ ◇ ◇ ◇

- 数値計算量を削減するため、 $p_T \rightarrow (B_t^2 + B_u^2)/2$ 、 $\mathbf{v} \cdot \mathbf{B} \rightarrow (v_t B_t + v_u B_u)$ とするなど、できるだけ代数的に整理している。
- if 文が計算が遅い場合は、sign 関数を用いてプログラムを書き換えることができる。

```

rhdli = 1.0_DP/(rhd1+0.5_DP-sign(0.5_DP,abs(rhd1)-eps))
rhdri = 1.0_DP/(rhdr+0.5_DP-sign(0.5_DP,abs(rhdr)-eps))

hl  = 0.5_DP+sign(0.5_DP,vnc)
hr  = 1.0_DP-hl
h2l = 0.5_DP+sign(0.5_DP,vnc-dabs(bnc)/rro2l)
h3l = (1.0_DP-h2l)*hl
h2r = 0.5_DP-sign(0.5_DP,vnc+dabs(bnc)/rro2r)
h3r = (1.0_DP-h2r)*hr

```

```

rou = ro2l*h1l+ro2r*hr
vsu = vs2l*h2l+vs3m*h3l+vs3m*h3r+vs2r*h2r
vtu = vt2l*h2l+vt3m*h3l+vt3m*h3r+vt2r*h2r
bsu = bs2l*h2l+bs3m*h3l+bs3m*h3r+bs2r*h2r
btu = bt2l*h2l+bt3m*h3l+bt3m*h3r+bt2r*h2r
enu = en2l*h2l+en3l*h3l+en3r*h3r+en2r*h2r

```

- MHD 方程式の流束は回転対称なので、上の 1 次元数値流束サブルーチンをそのまま多次元計算に利用できる。

```

...
vn = vy
vt = vz
vu = vx
...
call calc_flux_hlld(rol,vnl,vtl,vul,...
...
fmx_y = fmu
fmy_y = fmn
fmz_y = fmt
...

```

有限体積的なアプローチであれば、非構造格子も含むカーテシアン座標以外でも利用できる。